UPPSALA UNIVERSITET

PAGE 2005 – Pamplona, Spain

Challenges in modelling the pharmacokinetics of isoniazid in South African tuberculosis patients

Justin J Wilkins[1], Grant Langdon[1], Helen McIlleron[1], Goonaseelan Pillai[2], Peter J Smith[1] and Ulrika S H Simonsson[3]

(1) Division of Clinical Pharmacology, Department of Medicine, University of Cape Town, South Africa

(2) Modelling & Simulation, Clinical Development & Medical Affairs, Novartis Pharmaceuticals AG, Basel, Switzerland

(3) Division of Pharmacokinetics and Drug Therapy, Department of Pharmaceutical Biosciences, Uppsala University, Sweden

Isoniazid (INH)



- Key part of first-line tuberculosis chemotherapy
- Rapidly and completely absorbed from the intestine
- Subject to first-pass effect
- Substantial presystemic metabolism occurs in intestinal mucosal cells
- Subject to metabolic polymorphism – trimodal

UPPSALA UNIVERSITET

Patient Data

- 266 pulmonary tuberculosis (PTB) patients on first-line chemotherapy
- On treatment for 8-10 days prior to start of study
- 200-400 mg p.o. daily
- Samples for PK analysis were taken over between 1 and 6 weeks

UPPSALA UNIVERSITET

Representative Individual Profiles



UPPSALA UNIVERSITET

Modelling Procedure

- 2-compartment model with first-order absorption & elimination



- Model assessment criteria
  - ΔOFV
  - GOF plots
  - Scientific plausibility
  - Precision & accuracy

UPPSALA UNIVERSITET

Complex absorption

- Modelled using dual absorption compartments – fast and slow, characterized by absorption half-lives $HL_{a,fast}$ and $HL_{a,slow}$



$$F_{slow} = 1 - F_{fast}$$

```
HLFAST = THETA(3)
HLSLOW = THETA(4)+HLA1
```

UPPSALA UNIVERSITET

Other Aspects

IOV/IIV

- IOV/IIV characterization was a key objective
- 27 dosing occasions, IOV on 2 parameters – 56 ETA parameters needed

$$(CL/F)_i = \theta_{CL/F} \cdot \exp\left(\eta_i^{CL/F} + \kappa_{ij}^{CL/F}\right)$$

Karlsson MO, Sheiner LB. The importance of modeling interoccasion variability in population pharmacokinetic analyses. *J Pharmacokinet Biopharm*. 1993 Dec;**21**(6):735-50.

UPPSALA UNIVERSITET

Other Aspects

Acetylator phenotype

- Fast, intermediate and slow
- Data did not allow identification of intermediate group (indistinguishable from fast group)

```
$MIX
NSPOP = 2
P(1)  = THETA(8)
P(2)  = 1 - THETA(8)
...
```

UPPSALA UNIVERSITET

Other Aspects

Enterohepatic circulation

- Several models tested to account for enterohepatic circulation (EHC)
- EHC models were unable to produce either a substantial ΔOFV or a successful covariance step
- No literature evidence

Covariate relations

- GAM as implemented in Xpose used to identify potential covariate relations
- Screened using stepwise covariate modelling (SCM) method

$$TV(CL/F) = \theta_{CL/F} \cdot \left[1 + \theta_{WT} \cdot (WT - WT_{med})\right]$$

$$TV(V/F) = \theta_{V/F} + \theta_{SEX} \cdot SEX$$

UPPSALA UNIVERSITET

The Model



UPPSALA UNIVERSITET

Goodness-of-Fit









UPPSALA UNIVERSITET

Goodness-of-Fit



UPPSALA UNIVERSITET

Parameter Estimates

Parameter	Mean	% RSE	IIV (IOV)
Oral clearance (CL_{fast}/F, L.h^{-1})	10.4	2.90	0.156 (0.0327)
Oral clearance (CL_{slow}/F, L.h^{-1})	4.42	3.48	
Ratio of fast acetylators to slow acetylators (P_{CL})	0.225	14.9	
Half-life of absorption for fast compartment ($HL_{a,fast}$, h)	0.518	4.15	0.551 (0.351)
Half-life of absorption for slow compartment ($HL_{a,slow}$, h)	5.78	5.55	1.09
Dose fraction, fast cpt (F_{fast})	0.526	3.73	
Apparent volume of distribution, central compartment (V_1/F, L)	6.86	9.31	0.268
Apparent volume of distribution, peripheral compartment (V_2/F, L)	13.0	Fixed	
Intercompartmental rate constant (k_{12}, h^{-1})	2.43	12.2	
Intercompartmental rate constant (k_{21}, h^{-1})	1.75	8.51	
Residual Variability			
Constant coefficient of variability	0.191	3.03	

UPPSALA UNIVERSITET

The Final Model



SCM

Parsimonious Model for INH PK

UPPSALA UNIVERSITET

Acknowledgments

Aspects of this project were funded by the **Medical Research Council of South Africa.**

Grateful thanks to Jean van Dyk, Afia Fredericks and the patients and staff of **D P Marais SANTA Centre** and **Brewelskloof Hospital** for the input, support and cooperation they have lent this work.

UPPSALA UNIVERSITET

Complex Absorption



UPPSALA UNIVERSITET

Time = θ h θ_{GBL} h

Enterohepatic circulation



Adapted from the multiple-dose enterohepatic circulation model suggested by Luann Phillips (after Stuart Beal). NMusers, 11 June 2002. [http://www.cognigencorp.com/nonmem/nm/98may312002.html]

UPPSALA UNIVERSITET

Bimodal absorption

```
...
F2    = THETA(5)              ; F for fast comp
F3    = (1-F2)                ; F for slow comp
...
HLA1 = THETA(3)*EXP(BSV + BOV)
  ; abs half-life of fast abs comp
HLA2 = THETA(4)*EXP(ETA(4)) + HLA1
  ; abs half-life of slow abs comp
K21   = 0.693/HLA1
K31   = 0.693/HLA2
...
```

UPPSALA UNIVERSITET

Modelling Issues: IOV

```
...
$OMEGA BLOCK(1) .04
$OMEGA BLOCK(1) SAME
$OMEGA BLOCK(1) SAME
...

$PK
BSV = ETA(1)
BOV = ETA(5)
IF (OCC.EQ.2) BOV = ETA(6)
IF (OCC.EQ.3) BOV = ETA(7)
IF (OCC.EQ.4) BOV = ETA(8)
...
HLA1 = THETA(3)*EXP(BSV + BOV)
...
```

UPPSALA UNIVERSITET

Modelling Issues: Acetylator Phenotype

```
$MIX
NSPOP = 2
P(1)   = THETA(8)
P(2)   = 1 - THETA(8)
...
Q1 = 0
Q2 = 0
IF (MIXNUM.EQ.1) Q1 = 1
IF (MIXNUM.EQ.2) Q2 = 1

CL1   = THETA(1)
CL2   = THETA(9)

TVCL = ((CL1*Q1) + (CL2*Q2)) * (1+
   THETA(11)*(WT-50))
CL    = TVCL*EXP(BSV2 + BOV2)
...
```

UPPSALA UNIVERSITET

Full Final Model

```
$PROB INH FINAL
$DATA inh_dual_jun04.csv IGNORE=@
$INPUT ID IDNO=DROP OCCO=DROP DAY=DROP RATE=DROP
       TIME TT=DROP DV MDV AMT CMT EVID
       AGE SEX WT HT=DROP BMI=DROP
       RACE=DROP SMOK ALC PKG=DROP HIV
       HB=DROP HCT RBC=DROP MCV=DROP WBC=DROP
       AP=DROP ALT=DROP AST CRT=DROP
       TBIL=DROP UREA=DROP RIFP=DROP PZAP
       FDC DS=DROP
       LOC OCCD=DROP CLCR BSA=DROP OCC DRUG

$SUBROUTINE ADVAN6 TRANS1 TOL=5

$MODEL COMP = (CENT)
       COMP = (ABS1); fast abs comp
       COMP = (ABS2); slow abs comp
       COMP = (PERI)

...
```

```
...
$THETA (0, 4.6)          ;1  CL1
$THETA (0, 21.2)         ;2  V1
$THETA (0, 0.3)          ;3  HLA1
$THETA (0, 6.1)          ;4  HLA2
$THETA (0, .6, 1)        ;5  F-fast
$THETA (0 FIX)           ;6  ADD error
$THETA (0, 0.22)         ;7  CCV error
$THETA (0, 0.23, 1)      ;8  P1
$THETA (0, 11)           ;9  CL2
$THETA 0.01              ;10 SEX on V1
$THETA 0.01              ;11 WT on CL
$THETA 13 FIX            ;12 V2
$THETA (0, 2.5)          ;13 K14
$THETA (0, 1.8)          ;14 K41

$OMEGA .2 .3 .5 1.35

$OMEGA BLOCK(1) .04
$OMEGA BLOCK(1) SAME
$OMEGA BLOCK(1) SAME
...
```

UPPSALA UNIVERSITET

Full Final Model

```
... < many $OMEGA BLOCK(1) records omitted > ...

$OMEGA BLOCK(1) SAME
$OMEGA BLOCK(1) SAME

$SIGMA 1 FIX

$ABBREVIATED DERIV2=NO

$MIX
NSPOP = 2
P(1)  = THETA(8)
P(2)  = 1 - THETA(8)

$PK

BSV = ETA(1)
BOV = ETA(5)
IF (OCC.EQ.2)  BOV = ETA(6)
IF (OCC.EQ.3)  BOV = ETA(7)
IF (OCC.EQ.4)  BOV = ETA(8)
IF (OCC.EQ.5)  BOV = ETA(9)
IF (OCC.EQ.6)  BOV = ETA(10)
...
```

```
... < many $ETA statements omitted > ...

IF (OCC.EQ.26) BOV = ETA(30)
IF (OCC.EQ.27) BOV = ETA(31)

BSV2 = ETA(3)
BOV2 = ETA(32)
IF (OCC.EQ.2)  BOV2 = ETA(33)
IF (OCC.EQ.3)  BOV2 = ETA(34)

... < many $ETA statements omitted > ...

IF (OCC.EQ.26) BOV2 = ETA(57)
IF (OCC.EQ.27) BOV2 = ETA(58)

Q1 = 0
Q2 = 0

IF (MIXNUM.EQ.1) Q1 = 1
IF (MIXNUM.EQ.2) Q2 = 1

MXN=0
IF (MIXNUM.EQ.1) MXN=1
IF (MIXNUM.EQ.2) MXN=2
...
```

UPPSALA UNIVERSITET

Full Final Model

```
...

CL1   = THETA(1)                    ; CL1
CL2   = THETA(9)                    ; CL2

TVCL = ((CL1*Q1) + (CL2*Q2)) *
       (1+ THETA(11)*(WT-50))
CL    = TVCL*EXP(BSV+BOV)

TVV1 = THETA(2) + THETA(10)*SEX    ; V1
V1    = TVV1*EXP(ETA(2))            ; V1

; abs half-life of fast abs comp
HLA1 = THETA(3)*EXP(BSV2+BOV2)
; abs half-life of slow abs comp
HLA2 = THETA(4)*EXP(ETA(4))+HLA1
K21   = 0.693/HLA1
K31   = 0.693/HLA2

V2    = THETA(12)
K14   = THETA(13)
K41   = THETA(14)

F2    = THETA(5)            ; F for fast abs
F3    = (1-F2)              ; F for slow comp

S1    = V1
K10   = CL/V1
...
```

```
...

$ERROR

AA1    = A(1)
AA2    = A(2)
AA3    = A(3)
AA4    = A(4)
CP     = A(1)/V1

IPRED = A(1)/S1+0.00001
IRES  = DV-IPRED
W     = SQRT(THETA(6)**2 +
        THETA(7)**2*IPRED*IPRED)
IWRES = IRES/W
Y     = IPRED+W*EPS(1)

$DES

DADT(1) = K21*A(2) + K31*A(3) - K10*A(1) -
          K14*A(1) + K41*A(4)

DADT(2) = -K21*A(2)
DADT(3) = -K31*A(3)
DADT(4) = K14*A(1) - K41*A(4)

...
```

UPPSALA UNIVERSITET

Full Final Model

```
$EST POSTHOC NOABORT PRINT=10 MAXEVAL=9999
     MSFO=run526.msf SIGDIG=3

$COV PRINT=E MATRIX=S

$TABLE ID TIME IPRED IWRES ETA1 ETA2 ETA3
     ETA4 ETA5 ETA6
       ETA33 OCC AA1 AA2 AA3 AA4
       NOPRINT ONEHEADER FILE=sdtab526

$TABLE ID CL V1 HLA1 HLA2 F2 F3 K21 K31 K10
     K14 K41 V2 MXN
       NOPRINT ONEHEADER FILE=patab526

$TABLE ID AGE WT HCT AST
       NOPRINT ONEHEADER FILE=cotab526

$TABLE ID SEX SMOK ALC HIV PZAP FDC LOC
       NOPRINT ONEHEADER FILE=catab526

$TABLE ID AUC CP
       NOPRINT ONEHEADER FILE=run526.fit
```